Equational Abstractions

José Meseguer^1, Miguel Palomino^{1,2}, and Narciso Martí-Oliet^2

1 Computer Science Department, University of Illinois at Urbana-Champaign 2 Departamento de Sistemas Informáticos, Universidad Complutense de Madrid

meseguer@cs.uiuc.edu {miguelpt,narciso}@sip.ucm.es

Abstract. Abstraction reduces the problem of whether an infinite state system satisfies a temporal logic property to model checking that property on a finite state abstract version. The most common abstractions are quotients of the original system. We present a simple method of defining quotient abstractions by means of equations collapsing the set of states. Our method yields the minimal quotient system together with a set of proof obligations that guarantee its executability and can be discharged with tools such as those in the Maude formal environment.

1 Introduction

Abstraction techniques (see for example [1–15]) allow reducing the problem of whether an infinite state system, or a finite but too large one, satisfies a temporal logic property to model checking that property on a finite state abstract version. The most common way of defining such abstractions is by a quotient of the original system’s set of states, together with abstract versions of the transitions and the predicates. Many methods differ in their details but agree on their general use of a quotient map. There is always a minimal system (Kripke structure) making this quotient map a simulation. We present a simple method to build minimal quotient abstractions in an equational way. The method assumes that the concurrent system has been specified by means of a rewrite theory R = (Σ, E, R), with (Σ, E) an equational theory specifying the set of states as an algebraic data type, and R specifying the system’s transitions as a set of rewrite rules. The method consists on adding more equations, say E^′, to get a quotient system specified by the rewrite theory R/E^′ = (Σ, E ∪ E^′, R). We call such a system an equational abstraction of R. This equational abstraction is useful for model checking purposes if: (1) R/E^′ is an executable rewrite theory in a appropriate sense; and (2) the state predicates are preserved by the quotient simulation. Requirements (1) and (2) are proof obligations that can be discharged by theorem proving methods. Our approach can be mechanized using the rewriting logic language Maude [16] and its associated LTL model checker [17], inductive theorem prover [18], Church-Rosser checker [19], and coherence checker [20]. Our present experience with case studies, involving different abstractions discussed in the literature, suggests a fairly wide applicability for this method.

After summarizing LTL prerequisites (Sect. 2) and discussing simulations (Sect. 3), we explain in Sect. 4 how a concurrent system specified by a rewrite theory R has an associated Kripke structure giving semantics to its LTL properties; we also explain how Maude can model check such LTL properties for initial states having finitely many reachable states. Equational abstractions, their associated proof methods, and case studies are discussed in Sect. 5. Sect. 6 discusses related work and future research. Proofs of all the results in this paper and all case studies can be found in [21].

2 Prerequisites on Kripke Structures and LTL

To specify the properties of interest about our systems we will use linear temporal logic (LTL), which is interpreted in a standard way in Kripke structures. In what follows, we assume a fixed non-empty set of atomic propositions AP. Definition 1. A Kripke structure is a triple A = (A, →_A, L_A), where A is a set of states, →_A⊆ A × A is a total transition relation, and L_A : A → P(AP) is a labeling function associating to each state the set of atomic propositions that hold in it.

We will usually employ the notation a →_A b to say that (a, b) ∈ →_A. Note that the transition relation must be total, that is, for each a ∈ A there is a b ∈ A such that a →_A b. Given an arbitrary relation →, we write →^• for the total relation that extends → by adding a pair a →^• a for each a such that there is no b with a → b. A path in a Kripke structure A is a function π : IN −→ A such that, for each i ∈ IN, π(i) →_A π(i + 1).

The syntax of LTL(AP) is given by the following grammar: φ = p ∈ AP | φ ∨ φ | ¬φ | ◯ φ | φ U φ . The semantics of the logic, specifying the satisfaction relation A, a |= φ between a Kripke structure A, an initial state a ∈ A, and φ ∈ LTL(AP), is defined as usual (see for example [4, Sect. 3.1], where φ U ψ and ◯ φ are expressed in CTL^∗ notation as A(φUψ) and AXφ). Other Boolean and temporal operators (e.g., ⊤, ⊥, ∧, →, □, ◇, R, and ⊏) can be defined as syntactic sugar.

It is sometimes useful to restrict ourselves to the negation-free fragment LTL^−(AP) of LTL(AP), defined as follows: φ = p ∈ AP | ⊤ | ⊥ | φ ∨ φ | φ ∧ φ | ◯ φ | φ U φ | φ R φ . Negation is no longer available in LTL^−, and therefore the duals of the basic operators must be considered as basic ones, too. Since LTL^− is a sublogic of LTL, its semantics is the same. Furthermore, in a very practical sense there is no real loss of generality by restricting ourselves to formulas in LTL^−, because we can always transform any LTL formula φ into a semantically equivalent LTL^− formula φ̂. For that, we consider the extended set of atomic propositions ^AP = AP ∪ AP, where AP = {p̄ | p ∈ AP}, and construct φ̂ by first forming the negation normal form of φ (i.e., all negations are pushed to the atoms), and then replacing each negated atom ¬p by p̄. Given A = (A, →_A, L_A), we define ^A = (A, →_A, L ^A) where L ^A(a) = L^A(a) ∪ {p̄ ∈ AP | p /∈ L^A(p)}. Then we have, A, a |= φ ⇐⇒ ^A, a |= φ̂.

3 Simulations

We present a notion of simulation similar to that in [4], but somewhat more general (simulations in [4] essentially correspond to our strict simulations).

Definition 2. Given Kripke structures A = (A, →_A, L_A) and B = (B, →_B, L_B), both having the same set AP of atomic propositions, an AP -simulation H : A −→ B of A by B is given by a total binary relation H ⊆ A × B such that:

– if a →_A a^′ and aHb, then there is b^′ ∈ B such that b →_B b^′ and a^′Hb^′, and – if a ∈ A, b ∈ B, and aHb, then L_B(b) ⊆ L_A(a).

If the relation H is a function, then we call H an AP -simulation map. If both H and H^{−1} are AP -simulations, then we call H an AP -bisimulation. Also we call H strict if aHb implies L_B(b) = L_A(a).

The first condition guarantees that there is an abstract path in B corresponding to each concrete path in A; the second condition guarantees that an abstract state in B can satisfy only those atomic propositions that hold in all the concrete states in A that it simulates.

We say that an AP -simulation H : A −→ B reflects the satisfaction of an LTL formula φ ∈ LTL iff B, b |= φ and aHb imply A, a |= φ. The following theorem slightly generalizes Thm. 16 in [4]:

Theorem 1. AP -simulations always reflect satisfaction of LTL^−(AP) formulas. In addition, strict simulations also reflect satisfaction of LTL(AP) formulas.

This theorem is the key basis for the method of model checking by abstraction: given an infinite (or too large) system M, find a finitely reachable system A that simulates it and use model checking to try to prove that φ holds in A; then, by Thm. 1, φ also holds in M. In general, however, we typically only have our concrete system M and a surjective function h : M −→ A mapping concrete states to a simplified (usually finitely reachable) abstract domain A. In these cases there is a canonical way of constructing a Kripke structure out of h in such a way that h becomes a simulation.

Definition 3. The minimal system M^h_{min} corresponding to M and the surjective function h : M −→ A is given by the triple (A, h(→_M), L_Mh_{min} ), where L_Mh_{min} (a) = ⋂ x∈h^{−1}(a) LM(x). The following proposition is an immediate consequence of the definitions.

Proposition 1. For all such M and h, h : M −→ M^h_{min} is a simulation map.

Minimal systems can also be seen as quotients. Let A = (A, →_A, L_A) be a Kripke structure on AP , and let ≡ be an arbitrary equivalence relation on A. We can use ≡ to define a new Kripke structure, A/≡ = (A/≡, →_{A/≡}, L_{A/≡}), where:

– [a_1] →_{A/≡} [a_2] iff there exists a^′_1 ∈ [a_1] and a^′_2 ∈ [a_2] such that a^′_1 →_A a^′_2; – L_{A/≡}([a]) = ⋂ x∈[a] LA(x). It is then trivial to check that the projection map to equivalence classes q_≡ : a → [a] is an AP -simulation map q_≡ : A −→ A/≡, which we call the quotient abstraction defined by ≡. Hence, an equivalent presentation of the minimal system is expressed by the following.

Proposition 2. Let M = (M, →_M, L_M) be a Kripke structure and h : M −→ A a surjective function. Then, there exists a strict bijective bisimulation map between the Kripke structures M^h_{min} and M/≡_h, where by definition x ≡_h y iff h(x) = h(y). That is, we can perform the abstraction either by mapping the concrete states to an abstract domain or, as we will do in Sect. 5, by identifying some states and thereafter working with the corresponding equivalence classes.

The use of the adjective “minimal” is appropriate since, as pointed out in [3], M^h_{min} is the most accurate approximation to M that is consistent with h. However, it is not always possible to have a computable description of M^h_{min}. The definition of →_Mh_{min} can be rephrased as x →_Mh_{min} y iff there exist a and b such that h(a) = x, h(b) = y, and a →_M b. This relation, even if →_M is recursive, is in general only recursively enumerable. However, Sect. 5 develops equational methods that, when successful, yield a computable description of M^h_{min}.

4 Rewriting Logic Specifications and Model Checking

One can distinguish two specification levels: a system specification level, in which the computational system of interest is specified; and a property specification level, in which the relevant properties are specified. The main interest of rewriting logic [22] is that it provides a very flexible framework for the system-level specification of concurrent systems. A concurrent system is axiomatized by a rewrite theory R = (Σ, E, R), where (Σ, E) is an equational theory describing its set of states as the algebraic data type T_{Σ/E,k} associated to the initial algebra T_{Σ/E} of (Σ, E) by the choice of a type k of states in Σ^3. The system’s transitions are axiomatized by the conditional rewrite rules R which are of the form l : t −→ t^′ ⇐ cond, with l a label, t and t^′ Σ-terms, possibly with variables, and cond a condition^4. Under reasonable assumptions about E and R, rewrite theories are executable (more on this below). Indeed, there are several rewriting logic

3 We allow very general equational theories in membership equational logic [23], that can have types, subtypes defined by semantic conditions, and operator overloading. The desired set of states is then described by the carrier T_{Σ/E,k} of the initial algebra T_{Σ/E} for one of those types k, technically called either sorts or kinds in [23]. The elements of T_{Σ/E} are E-equivalence classes of terms [t]E ; that is, two terms are equal iff they can be proved so by E.

4 In this paper we assume that the condition cond can involve a conjunction of equations u = v and memberships of the form w : s stating that the term w has sort s. The conjunction must hold for a substitution instance θ before we are allowed to rewrite θ(t) to θ(t^′). We also assume that vars(t^′) ∪ vars(cond) ⊆ vars(t).

language implementations, including ELAN [24], CafeOBJ [25], and Maude [16]. We can illustrate rewriting logic specifications by means of a simple example, namely Lamport’s bakery protocol [26]. This is an infinite state protocol that achieves mutual exclusion between processes by dispensing a number to each process and serving them in sequential order according to the number they hold. A simple Maude specification for the case of two processes is as follows:

mod BAKERY is protecting NAT .
  sorts Mode State .
  ops sleep wait crit : -> Mode .
  op <_,_,_,_> : Mode Nat Mode Nat -> State .
  op initial : -> State .
  vars P Q : Mode . vars X Y : Nat .
  eq initial = < sleep, 0, sleep, 0 > .
  rl [p1_sleep] : < sleep, X, Q, Y > => < wait, s Y, Q, Y > .
  rl [p1_wait] : < wait, X, Q, 0 > => < crit, X, Q, 0 > .
  crl [p1_wait] : < wait, X, Q, Y > => < crit, X, Q, Y > if not (Y < X) .
  rl [p1_crit] : < crit, X, Q, Y > => < sleep, 0, Q, Y > .
  rl [p2_sleep] : < P, X, sleep, Y > => < P, X, wait, s X > .
  rl [p2_wait] : < P, 0, wait, Y > => < P, 0, crit, Y > .
  crl [p2_wait] : < P, X, wait, Y > => < P, X, crit, Y > if Y < X .
  rl [p2_crit] : < P, X, crit, Y > => < P, X, sleep, 0 > .
endm

This specification corresponds to a rewrite theory R = (Σ, E, R), where (Σ, E) imports the equational theory NAT of the natural numbers, and where Σ has additional types (called here sorts) Mode and State, with Mode consisting of just the constants sleep, wait, and crit. States are represented by terms of sort State, which are constructed by a 4-tuple operator <_,_,_,_> ; the first two components describe the status of the first process (the mode it is currently in, and its priority as given by the number according to which it will be served), and the last two components the status of the second process. E consists of just the equations imported from NAT, plus the above equation defining the initial state. R consists of eight rewrite rules, four for each process. These rules describe how each process passes from being sleeping to waiting, from waiting to its critical section, and then back to sleeping. In this case, the chosen type k for states is of course State.

Rewriting logic then has inference rules to infer all the possible concurrent computations in a system [22], in the sense that, given two states [u], [v] ∈ T_{Σ/E,k}, we can reach [v] from [u] by some possibly complex concurrent computation iff we can prove R ⊢ u −→ v in the logic. In particular we can easily define the one-step R-rewriting relation, which is a binary relation →^1_{R,k} on T_{Σ,k} that holds between terms u, v ∈ T_{Σ,k} iff there is a one-step proof of R ⊢ u −→ v, that is, a proof in which only one rewrite rule in R is applied to a single subterm. We can get a binary relation (with the same name) →^1_{R,k} on T_{Σ/E,k} by defining [u] →^1_{R,k} [v] iff u^′ →^1_{R,k} v^′ for some u^′ ∈ [u], v^′ ∈ [v].

The relationship with Kripke structures is now almost obvious, since we can associate to a concurrent system axiomatized by a rewrite theory R = (Σ, E, R) with a chosen type k of states a Kripke structure, K(R, k)_Π = (T_{Σ/E,k}, (→^1_{R,k})^•, L_Π ). We say “almost obvious,” because nothing has yet been said about the choice of state predicates Π and the associated labeling function L_Π . The reason for this is methodological: Π, L_Π , and the LTL formulas φ describing properties of the system specified by R belong to the property specification level. Indeed, for the same system specification R we may come up with different predicates Π, labeling functions L_Π , and properties φ at the property specification level.

The question of when a rewrite theory R is executable is closely related with wanting T_{Σ/E,k} to be a computable set, and (→^1_{R,k})^• to be a computable relation in the above Kripke structure K(R, k)_Π , an obvious precondition for any model checking. We say that R = (Σ, E ∪A, R) is executable if: (1) there exists a matching algorithm modulo the equational axioms A^5; (2) the equational theory (Σ, E ∪ A) is (ground) Church-Rosser and terminating modulo A [27]; and (3) the rules R are (ground) coherent [28] relative to the equations E modulo A. Conditions (1–2) ensure that T_{Σ/E∪A,k} is a computable set, since each ground term t can be simplified by applying the equations E from left to right modulo A to reach a canonical form can_{E/A}(t) which is unique modulo the axioms A. We can then reduce the equality problem [u]_{E∪A} = [v]_{E∪A} to the decidable equality problem [can_{E/A}(u)]_A = [can_{E/A}(v)]_A. Condition (3) means that for each ground term t, whenever we have t →^1_R u we can always find can_{E/A}(t) →^1_R v such that [can_{E/A}(u)]_A = [can_{E/A}(v)]_A. This implies that (→^1_{R,k})^• is a computable binary relation on T_{Σ/E∪A,k}, since we can decide [t]_{E∪A} →^1_R [u]_{E∪A} by enumerating the finite set of all one-step R-rewrites modulo A of can_{E/A}(t), and for any such rewrite, say v, we can decide [can_{E/A}(u)]_A = [can_{E/A}(v)]_A.

4.1 LTL Properties of Rewrite Theories and Model Checking

One appealing feature of rewriting logic is that it provides a seamless integration of the system specification level and the property specification level, because we can specify the relevant state predicates Π equationally, and this then determines the labeling function L_Π and the semantics of the LTL formulas φ in a unique way. Indeed, to associate LTL properties to a rewrite theory R = (Σ, E ∪ A, R) with a chosen type k of states we only need to make explicit the relevant state predicates Π, which need not be part of the system specification R. The state predicates Π can be defined by means of equations D in an equational theory (Σ^′, E ∪ A ∪ D) extending (Σ, E ∪ A) in a conservative way; specifically, the unique Σ-homomorphism T_{Σ/E∪A} → T_Σ′_{/E∪A∪D} should be bijective at each sort s in Σ. The syntax defining the state predicates consists of a subsignature Π ⊆ Σ^′ of function symbols p of the general form p : s_1 . . . s_n −→ Prop (with Prop a shorthand for Proposition), reflecting the fact that state predicates can be parametric. The semantics of the state predicates Π is defined by D with the help of an operator |= : k Prop −→ Bool in Σ^′. By definition, given ground

5 In Maude, the axioms A for which the rewrite engine supports matching modulo are any combination of associativity, commutativity, and identity axioms for different binary operators.

terms u_1, . . . , u_n, we say that the state predicate p(u_1, . . . , u_n) holds in the state [t] iff E ∪ A ∪ D ⊢ t |= p(u_1, . . . , u_n) = true . We can now associate to R a Kripke structure K(R, k)_Π , whose atomic predicates are specified by the set AP_Π = {θ(p) | p ∈ Π, θ ground substitution}^6. We define K(R, k)_Π = (T_{Σ/E,k}, (→^1_{R,k})^•, L_Π ), where L_Π ([t]) = {θ(p) ∈ AP_Π | θ(p) holds in [t]}. In practice we want the equality t |= p(u_1, . . . , u_n) = true to be decidable. This can be achieved by giving equations in D ∪ E that are Church-Rosser and terminating modulo A. Then, if we begin with an executable rewrite theory R and define decidable state predicates Π by the method just described, we obtain a computable Kripke structure K(R, k)_Π which, if it has finite reachability sets, can be used for model checking.

The Maude 2.0 system has an on-the-fly, explicit-state LTL model checker [17] which supports the methodology just mentioned. Given an executable rewrite theory specified in Maude by a module M, and an initial state init of sort StateM, we can model check different LTL properties beginning at this state.

For that, a new module CHECK-M must be defined importing M and the predefined module MODEL-CHECKER, and a subsort declaration StateM < State must be added. Then the syntax of the state predicates must be declared by means of operations of sort Prop, and their semantics must be given by equations involving the satisfaction operator op _|=_ : State Prop -> Bool. Once the semantics of the state predicates has been defined, and assuming that the set of states reachable from init is finite, we can model check any LTL formula in LTL(AP_Π ) by giving to Maude the command: reduce modelCheck(init, formula).

Continuing with our bakery protocol example, two basic properties that we may wish to verify are: (1) mutual exclusion: the two processes are never simultaneously in their critical section; and (2) liveness: any process in waiting mode will eventually enter its critical section. In order to specify these properties it is enough to specify in Maude the following set Π of state predicates:

mod BAKERY-CHECK is inc MODEL-CHECKER . inc BAKERY .
  ops 1wait 2wait 1crit 2crit : -> Prop .
  vars P Q : Mode . vars X Y : Nat .
  eq (< P, X, Q, Y > |= 1wait) = (P == wait) .
  eq (< P, X, Q, Y > |= 2wait) = (Q == wait) .
  eq (< P, X, Q, Y > |= 1crit) = (P == crit) .
  eq (< P, X, Q, Y > |= 2crit) = (Q == crit) .
endm

Since the set of states reachable from initial (defined in the BAKERY module) is infinite, we should not model check the above specification as given. Instead, we should first define an abstraction of it where initial has only finitely many reachable states and then model check the abstraction.

6 By convention, if p has n parameters, θ(p) denotes the term θ(p(x1, . . . , xn)).

5 Equational Abstractions

Let R = (Σ, E ∪ A, R) be a rewrite theory. A quite general method for defining abstractions of the Kripke structure K(R, k)_Π = (T_{Σ/E∪A,k}, (→^1_{R,k})^•, L_Π ) is by specifying an equational theory extension of the form (Σ, E ∪ A) ⊆ (Σ, E ∪ A ∪ E^′). Since this defines an equivalence relation ≡_E′ on T_{Σ/E∪A,k}, namely, [t]_{E∪A} ≡_E′ [t^′]_{E∪A} ⇐⇒ E ∪ A ∪ E^′ ⊢ t = t^′ ⇐⇒ [t]_{E∪A∪E}′ = [t^′]_{E∪A∪E}′ , we can obviously define our quotient abstraction as K(R, k)_Π /≡_E′ . We call this the equational quotient abstraction of K(R, k)_Π defined by E^′.

But can K(R, k)_Π /≡_E′ , which we have just defined in terms of the underlying Kripke structure K(R, k)_Π , be understood as the Kripke structure associated to another rewrite theory? Let us take a closer look at K(R, k)_Π /≡_E′ = (T_{Σ/E∪A,k}/≡_E′ , (→^1_{R,k})^•/≡_E′ , L_{Π/≡E′} ). The first observation is that, by definition, we have T_{Σ/E∪A,k}/≡_E′ ∼= T_{Σ/E∪A∪E}′_{,k}. A second observation is that if R is k-deadlock free, that is, if we have (→^1_{R,k})^• = →^1_{R,k}, then the rewrite theory R/E^′ = (Σ, E ∪A∪E^′, R) is also k-deadlock free, and we have (→^1_{R/E}′_{,k})^• = →^1_{R/E}′_{,k} = (→^1_{R,k})^•/≡_E′ .

Therefore, for R k-deadlock free, our obvious candidate for a rewrite theory having K(R, k)_Π /≡_E′ as its underlying Kripke structure is the rewrite theory R/E^′ = (Σ, E ∪ A ∪ E^′, R). That is, we just add to R the equations E^′ and do not change at all the rules R. How restrictive is the requirement that R is k-deadlock free? There is no essential loss of generality: in Sect. 5.2 we show how we can always associate to an executable rewrite theory R a semantically equivalent (from the LTL point of view) theory R_{d.f.} which is both deadlock free and executable.

Therefore, at a purely mathematical level, R/E^′ seems to be what we want. Assuming that we have an A-matching algorithm, the problem comes with the following two executability questions about R/E^′, which are essential for K(R, k)_Π /≡_E′ to be computable, and therefore for model checking:

– Are the equations E ∪E^′ ground Church-Rosser and terminating modulo A? – Are the rules R ground coherent relative to E ∪ E^′ modulo A?

The answer to each of these questions may be positive or negative. In practice, sufficient care on the part of the user when specifying E^′ should result in an affirmative answer to the first question. In any case, we can always try to check such a property with a tool such as Maude’s Church-Rosser checker [19]; if the check fails, we can try to complete the equations with a Knuth-Bendix completion tool, for example [29, 30], to get a theory (Σ, E^{′′} ∪ A) equivalent to

(Σ, E ∪ A ∪ E^′) for which the first question has an affirmative answer. Likewise, we can try to check whether the rules R are ground coherent relative to E ∪ E^′ (or to E^{′′}) modulo A using the tools described in [20]. If the check fails we can again try to complete the rules R to a semantically equivalent set of rules R^′, using also those tools [20]. By this process we can hopefully arrive at an executable rewrite theory R^′ = (Σ, E^{′′} ∪ A, R^′) which is semantically equivalent to R/E^′. We can then use R^′ to try to model check properties about R.

But we are not finished yet. What about the state predicates Π? Recall (see the end of Sect. 4) that these (possibly parameterized) state predicates will have been defined by means of equations D in a Maude module importing the specification of R and also the MODEL-CHECKER module. The question is whether the state predicates Π are preserved under the equations E^′. This indeed may be a problem. We need to unpack a little the definition of the innocent-looking labeling function L_{Π/≡E′} , which is defined by the intersection formula L_{Π/≡E′} ([t]_{E∪A∪E}′ ) = ⋂ [x]_{E∪A}⊆[t]_{E∪A∪E}′ L_Π ([x]_{E∪A}).

In general, computing such an intersection and coming up with new equational definitions D^′ capturing the new labeling function L_{Π/≡E′} may not be easy. It becomes much easier if the state predicates Π are preserved under the equations E^′. By definition, we say that the state predicates Π are preserved under the equations E^′ if for any [t]_{E∪A}, [t^′]_{E∪A} ∈ T_{Σ/E∪A,k} we have the implication [t]_{E∪A∪E}′ = [t^′]_{E∪A∪E}′ =⇒ L_Π ([t]_{E∪A}) = L_Π ([t^′]_{E∪A}). Note that in this case, assuming that the equations E ∪ E^′ ∪ D (or E^{′′} ∪ D) are ground Church-Rosser and terminating modulo A, we do not need to change the equations D to define the state predicates Π on R/E^′ (or its semantically equivalent R^′). Therefore, we have an isomorphism (given by a pair of invertible bisimulation maps) K(R, k)_Π /≡_E′ ∼= K(R/E^′, k)_Π , or, in case we need the semantically equivalent R^′, an isomorphism K(R, k)_Π /≡_E′ ∼= K(R^′, k)_Π . The crucial point in both isomorphisms is that the labeling function of the righthand side Kripke structure is now equationally defined by the same equations D as before. Since by construction either R/E^′ or R^′ are executable theories, for an initial state [t]_{E∪A∪E}′ having a finite set of reachable states we can use the Maude model checker to model check any LTL formula in this equational quotient abstraction. Furthermore, since the quotient AP_Π -simulation map K(R, k)_Π −→ K(R/E^′, k)_Π is then by construction strict, it reflects satisfaction of arbitrary LTL formulas by Thm. 1.

A practical problem remains: how can we actually try to prove the implication [t]_{E∪A∪E}′ = [t^′]_{E∪A∪E}′ =⇒ L_Π ([t]_{E∪A}) = L_Π ([t^′]_{E∪A}) to show the desired preservation of state predicates? A particularly easy case is that of k-topmost rewrite theories, that is, theories in which the type k of states only appears as the codomain of an operation f : k_1 . . . k_n −→ k. This is not a very restrictive condition, since any rewrite theory R can be transformed into a semantically equivalent k^′-topmost one just by encapsulating the original type of states k in a new type k^′ through an operation { } : k −→ k^′ [21].

Proposition 3. Suppose a k-topmost rewrite theory in which all (possibly conditional) equations in E^′ are of the form t = t^′ if C with t, t^′ ∈ T_{Σ,k}, and that the equations E∪E^′ ∪D are Church-Rosser and terminating modulo A. Furthermore, suppose that no equations between terms in T_{Σ,k} appear in the conditions of any equation in E^′. If for each equation t = t^′ if C in E^′ and each state predicate p we can prove the inductive property E ∪ A ∪ D ⊢_{ind} (∀⃗x ∀⃗y) C → (t(⃗x) |= p(⃗y) = true ↔ t^′(⃗x) |= p(⃗y) = true) then we have established the preservation of the state predicates Π by the equations E^′.

We can use a tool like Maude’s ITP [18] to mechanically discharge proof obligations of this kind.

5.1 Case Studies

The Bakery Protocol Example Revisited. We can use the bakery protocol example to illustrate how equational quotient abstractions can be used to verify infinite-state systems. We can define such an abstraction by adding to the equations of BAKERY-CHECK a set E^′ of additional equations defining a quotient of the set of states. We can do so in the following module extending BAKERY-CHECK by equations and leaving the transition rewrite rules unchanged:

mod ABSTRACT-BAKERY-CHECK is inc BAKERY-CHECK .
  vars P Q : Mode . vars X Y : Nat .
  eq < P, 0, Q, s s Y > = < P, 0, Q, s 0 > .
  eq < P, s s X, Q, 0 > = < P, s 0, Q, 0 > .
  ceq < P, s X, Q, s Y > = < P, s s 0, Q, s 0 >
      if (Y < X) /\ not(Y == 0 and X == s 0) .
  ceq < P, s X, Q, s Y > = < P, s 0, Q, s 0 >
      if not (Y < X) /\ not (Y == 0 and X == 0) .
endm

Note that ⟨P, N, Q, M⟩ ≡ ⟨P^′, N^′, Q^′, M^′⟩ according to the above equations iff (1) P = P^′ and Q = Q^′, (2) N = 0 iff N^′ = 0, (3) M = 0 iff M^′ = 0, (4) M < N iff M^′ < N^′. Three key questions are: (1) Is the set of states now finite?; (2)

Does this abstraction correspond to a rewrite theory whose equations are ground Church-Rosser and terminating?; (3) Are the rules still ground coherent?

The equations are indeed ground Church-Rosser and terminating. It is also clear that the set of states is now finite, since in the canonical forms obtained with these equations the natural numbers possible in the state can never be greater than s(s(0)). This leaves us with the ground coherence question. We have to analyze possible “relative critical pairs” between rules and equations. For example, consider the following pair of a rule and an equation:

rl [p1_sleep] : < sleep, X, Q, Y > => < wait, s Y, Q, Y > .
eq < P, 0, Q, s s Y > = < P, 0, Q, s 0 > .

The only possible overlap corresponds to the unification (after making the variables disjoint) of the two lefthand sides yielding the term < sleep, 0, Q, s s Y >, which is rewritten by the rule to < wait, s s s Y, Q, s s Y > and by the equation to < sleep, 0, Q, s 0 >, with both terms being finally reduced to < wait, s s 0, Q, s 0 >, in the first case by means of the third equation, and in the second one by the rule [p1 sleep]. All the other rule-equation pairs can likewise be proved coherent. A fourth pending question is the deadlock freedom of BAKERY-CHECK. This property holds and can be checked with the ITP.

What about state predicates? Are they preserved by the abstraction? Note that, since the rewrite theory is State-topmost and the equations are all between terms of sort State, according to Prop. 3 we only need to check that each of the equations preserves the above state predicates. But this is trivial, since the predicates only depend on Mode components that are left unchanged by the equations. This can be mechanically checked using Maude’s ITP [21].

In other words, we have just shown that, for Π the state predicates declared in BAKERY-CHECK, we have a strict quotient simulation map,

K(BAKERY-CHECK, State)_Π −→ K(ABSTRACT-BAKERY-CHECK, State)_Π .

Therefore, we can establish the mutual exclusion property of BAKERY-CHECK by model checking in ABSTRACT-BAKERY-CHECK the following:

reduce modelCheck(initial, []~ (1crit /\ 2crit)) .
result Bool: true

Likewise, we can establish the liveness property of BAKERY-CHECK by model checking in ABSTRACT-BAKERY-CHECK:

reduce modelCheck(initial, (1wait |-> 1crit) /\ (2wait |-> 2crit)) .
result Bool: true

Other Examples. In addition to the bakery protocol we have also dealt successfully with a number of examples that have been used in the literature to illustrate other abstraction methods, including a readers/writers system [11], the alternating bit protocol [13, 6, 12], a mutual exclusion protocol discussed in

[7], and the bounded retransmission protocol [1, 2, 6]. The abstractions were obtained simply by adding some equations to the specifications. Only in the last two cases was it necessary to add some extra (but semantically equivalent) rules to guarantee coherence; the details can be found in [21].

5.2 The Deadlock Difficulty

The reason why we have focused on deadlock-free rewrite theories is because deadlocks can pose a problem due to a subtle point in the semantics of LTL. As emphasized in its definition, the transition relation of a Kripke structure is total, and this requirement is also imposed on the Kripke structures arising from rewrite theories. Consider then the following specification of a rewrite theory, together with the declaration of two state predicates:

mod FOO is inc MODEL-CHECKER .
  ops a b c : -> State . ops p1 p2 : -> Prop .
  eq (a |= p1) = true . eq (b |= p2) = true . eq (c |= p1) = true .
  rl a => b . rl b => c .
endm

The transition relation of the Kripke structure corresponding to this specification has three elements: a → b, b → c, and c → c, the last one consistently added by the model checker according to the semantics given to LTL. Suppose now that we wanted to abstract this system and that we decided to identify a and c by means of a simulation map h. For that, according to the previous sections, it would be enough to add the equation eq c = a to the above specification. The resulting system is coherent, and a and c satisfy the same state predicates. Note that the resulting Kripke structure has only two elements in its transition relation: one from the equivalence class of a to that of b, and another in the opposite direction. Now, since no deadlock can occur in any of the states, the model checker does not add any additional transition steps. In particular, there is no transition from the equivalence class of a to itself, but that means that the resulting specification does not correspond to the minimal system associated to h in which such a transition does exist. The lack of this idle transition is a serious problem, because now we can prove properties about the simulating system that are actually false in the original one, for example, □ ◇ p2.

One simple way to deal with this difficulty is to just add idle transitions for each of the states in the resulting specification by means of a rule of the form x => x. The resulting system, in addition to all the rules that the minimal system should contain, may in fact have some extra “junk” rules that are not part of it. Therefore, we end up with a system that can be soundly used to infer properties of the original system (it is immediate to see that we have a simulation map) but that in general will be coarser than the minimal system.

A better way of addressing the problem is to characterize the set of deadlock states. For this, given a rewrite theory R we can define a new operation enabled : k −→ Bool? for each type k in R, where Bool? is a supersort of Bool. Then we add, for each rule t → t^′ if C, the equation enabled(t) = true if C and, for each operation f : k_1 . . . k_n −→ k, n equations of the form enabled(f(x_1, . . . , x_n)) = true if enabled(x_i) = true, so that (∃t^′) t →^1_{R,k} t^′ iff enabled(t) = true. This enabled predicate is the key point in the proof of the following proposition, which allows us to transform an executable rewrite theory into a semantically equivalent one that is both deadlock-free and executable.

Proposition 4. Let R = (Σ, E ∪ A, R) be an executable rewrite theory. Given a chosen type of states k, we can construct an executable theory extension R ⊆ R^k_{d.f.} = (Σ^′, E^′ ∪ A, R^′) such that:

– R^k_{d.f.} is k^′-deadlock free and k^′-topmost for a certain type k^′; – there is a function h : T_Σ′_{,k}′ −→ T_{Σ,k} inducing a bijection h : T_Σ′_{/E}′_{∪A,k}′ −→ T_{Σ/E∪A,k} such that for each t, t^′ ∈ T_Σ′_{,k}′ we have h(t)(→^1_{R,k})^•h(t^′) ⇐⇒ t →^1_Rk_{d.f.,k′} t^′. Furthermore, if Π are state predicates for R and k defined by equations D, then we can define state predicates Π for R^k_{d.f.} and k^′ by equations D^′ such that the above map h becomes a bijective AP_Π -bisimulation h : K(R^k_{d.f.}, k^′)_Π −→ K(R, k)_Π . 6 Related Work and Conclusions

In [3] the simulation of a system M by another M^′ through a surjective function h was defined and the optimal simulation M^h_{min} was identified. The idea of simulating by quotient has been further explored in [4, 5, 2, 10, 12, 7] among others, although the construction in [7] requires a Galois connection instead of just a function. Theorem proving is proposed in [2] to construct the transition relation of the abstract system, and in [12] to prove that a function is a representative function that can be used as input to an algorithm to extract M^h_{min} out of M. While those uses of theorem proving focus on the correctness of the abstract transition relation, our method focuses on making the minimal transition relation (which is correct by construction) computable, and on proving the preservation of the labeling function. In [3, 4], on the other hand, the minimal model M^h_{min} is discarded in favor of less precise but easier to compute approximations; this would correspond, in our approach, to the addition of rewrite rules to the specification to simplify the proofs of the proof obligations. In all the papers mentioned two states can become identified only if they satisfy the same atomic propositions; our definition of simulation is more general, but we have not yet exploited this.

The equational abstraction method that we have presented seems to apply in practice to a good number of examples discussed in the literature. But we need to further test its applicability on a wider and more challenging range of examples. Also, the method itself should be generalized along several directions.

For example, we should generalize the equational theory extension (Σ, E ∪ A) ⊆ (Σ, E ∪A∪E^′ ) to an arbitrary theory interpretation H : (Σ, E ∪A) −→ (Σ^′, E^{′′}), allowing arbitrary transformations on the data representation of states. A particular instance of this seems to be predicate abstraction [14, 6]. Under this approach, the abstract domain is a Boolean algebra over a set of assertions and the abstraction function, typically as part of a Galois connection, is symbolically constructed as the conjunction of all expressions satisfying a certain condition, which is proved using theorem proving. This would correspond to a theory interpretation H : (Σ, E) −→ (Σ ∪ Σ^′, E ∪ E^′), with Σ^′ introducing operations of the form p : State −→ Bool, and with H mapping states S to Boolean tuples ⟨p_1(S), . . . , p_n(S)⟩. Similarly, we should consider simulation maps between different sets AP and AP^′ of state predicates, yielding another increase in generality when relating systems. Finally, equational abstractions that do not require strict preservation of state predicates should also be investigated.

Acknowledgments. Research supported by ONR Grant N00014-02-1-0715, NSF Grant CCR-0234524, and by DARPA through Air Force Research Laboratory Contract F30602-02-C-0130; and by the Spanish project AMEVA CICYT TIC 2000–0701–C02–01. We warmly thank Saddek Bensalem, Yassine Lakhnech, David Basin, Felix Klaedtke, Natarajan Shankar, Hassen Saidi, and Tomás Uribe for many useful discussions that have influenced the ideas presented here, Manuel Clavel and Francisco Durán for their help in the preparation of this paper, and Roberto Bruni and Joe Hendrix for many useful comments on previous drafts.

References

1. Abdulla, P., Annichini, A., Bouajjani, A.: Symbolic verification of lossy channel systems: Application to the bounded retransmission protocol. In Cleaveland, W.R., ed.: Tools and Algorithms for the Construction of Analysis of Systems, TACAS’99. LNCS 1579., Springer (1999) 2. Bensalem, S., Lakhnech, Y., Owre, S.: Computing abstractions of infinite state systems compositionally and automatically. In Hu, A.J., Vardi, M.Y., eds.: Computer Aided Verification, CAV’98. LNCS 1427., Springer (1998) 319–331 3. Clarke, E.M., Grumberg, O., Long, D.E.: Model checking and abstraction. ACM Transactions on Programming Languages and Systems 16 (1994) 1512–1542 4. Clarke, E.M., Grumberg, O., Peled, D.A.: Model Checking. MIT Press (1999) 5. Clarke, E.M., Grumberg, O., Jha, S., Lu, Y., Veith, H.: Counterexample-guided abstraction refinement. In Emerson, E.A., Sistla, A.P., eds.: Computer Aided Verification, CAV’00. LNCS 1855., Springer (2000) 154–169 6. Colón, M.A., Uribe, T.E.: Generating finite-state abstractions of reactive systems using decision procedures. In Hu, A.J., Vardi, M.Y., eds.: Computer Aided Verification, CAV’98. LNCS 1427., Springer (1998) 293–304 7. Dams, D., Gerth, R., Grumberg, O.: Abstract interpretation of reactive systems.

ACM Transactions on Programming Languages and Systems 19 (1997) 253–291 8. Havelund, K., Shankar, N.: Experiments in theorem proving and model checking for protocol verification. In Gaudel, M.C., Woodcock, J., eds.: FME ’96: Industrial Benefit and Advances in Formal Methods. LNCS 1051., Springer (1996) 662–681

9. Kesten, Y., Pnueli, A.: Control and data abstraction: The cornerstones of practical formal verification. International Journal on Software Tools for Technology Transfer 4 (2000) 328–342 10. Kesten, Y., Pnueli, A.: Verification by augmentary finitary abstraction. Information and Computation 163 (2000) 203–243 11. Loiseaux, C., Graf, S., Sifakis, J., Bouajjani, A., Bensalem, S.: Property preserving abstractions for the verification of concurrent systems. Formal Methods in System Design 6 (1995) 1–36 12. Manolios, P.: Mechanical Verification of Reactive Systems. PhD thesis, Univ. of Texas at Austin (2001) 13. Müller, O., Nipkow, T.: Combining model checking and deduction for I/Oautomata. In Brinksma, E., et al., eds.: Tools and Algorithms for the Construction and Analysis of Systems, TACAS ’95. LNCS 1019., Springer (1995) 1–16 14. Saïdi, H., Shankar, N.: Abstract and model check while you prove. In Halbwachs, N., Peled, D., eds.: Computer Aided Verification, CAV’99. LNCS 1633., Springer (1999) 443–454 15. Uribe Restrepo, T.E.: Abstraction-Based Deductive-Algorithmic Verification of Reactive Systems. PhD thesis, Dept. of Computer Science, Stanford Univ. (1998) 16. Clavel, M., Durán, F., Ecker, S., Lincoln, P., Martí-Oliet, N., Meseguer, J., Quesada, J.F.: Maude: Specification and programming in rewriting logic. Theoretical Computer Science 285 (2002) 187–243 17. Eker, S., Meseguer, J., Sridharanarayanan, A.: The Maude LTL model checker. In Gadducci, F., Montanari, U., eds.: Rewriting Logic and its Applications, WRLA 2004. ENTCS 71., Elsevier (2002) 18. Clavel, M.: The ITP tool. In Nepomuceno, A., et al., eds.: Logic, Language, and Information, Kronos (2001) 55–62 19. Durán, F., Meseguer, J.: A Church-Rosser checker tool for Maude equational specifications. http://maude.cs.uiuc.edu/tools (2000) 20. Durán, F.: Coherence checker and completion tools for Maude specifications. http:

//maude.cs.uiuc.edu/tools (2000)

21. Meseguer, J., Palomino, M., Martí-Oliet, N.: Notes on model checking and abstraction in rewriting logic. http://formal.cs.uiuc.edu/texts/nmcarl.ps (2002) 22. Meseguer, J.: Conditional rewriting logic as a unified model of concurrency. Theoretical Computer Science 96 (1992) 73–155 23. Meseguer, J.: Membership algebra as a logical framework for equational specification. In Parisi-Presicce, F., ed.: Recent Trends in Algebraic Development Techniques WADT’97. LNCS 1376., Springer (1998) 18–61 24. Borovanský, P., Kirchner, C., Kirchner, H., Moreau, P.E.: ELAN from a rewriting logic point of view. Theoretical Computer Science 285 (2002) 155–185 25. Futatsugi, K., Diaconescu, R.: CafeOBJ Report. World Scientific (1998) 26. Lamport, L.: A new solution of Dijkstra’s concurrent programming problem. Communications of the ACM 17 (1974) 453–455 27. Dershowitz, N., Jouannaud, J.P.: Rewrite systems. In van Leeuwen, J., ed.: Handbook of Theoretical Computer Science, Vol. B. North-Holland (1990) 243–320 28. Viry, P.: Equational rules for rewriting logic. Theoretical Computer Science 285 (2002) 29. Contejean, E., Marché, C.: The CiME system: tutorial and user’s manual.

Manuscript, Univ. Paris-Sud, Centre d’Orsay 30. Durán, F.: Termination checker and Knuth-Bendix completion tools for Maude equational specifications. Manuscript, Computer Science Laboratory, SRI International, http://maude.cs.uiuc.edu/papers (2000)